Optimizing the Design of a Fuel Pod with NX and Maple

Introduction

A manufacturer has designed a fuel pod in NX. The fuel pod has a hemispherical and conical end, and a cylindrical midsection.



To minimize material costs, the manufacturer wants to minimize the surface area of the fuel pod while maintaining the existing volume.

This application

- pulls the current dimensions of the fuel pod (radius of the hemispherical end, length of the cylindrical midsection, and height of the conical end) from the NX CAD model,
- calculates the current volume of the fuel pod,
- optimizes the dimensions to minimize the surface area while maintaining the existing volume,
- and pushes the optimized dimensions back into the NX CAD model.

NOTE: To use this application, you must

- have a supported version of NX installed,
- load canisterOptimization.prt in NX (this is the CAD model of the fuel pod),
- ensure the NX-Maple link is working correctly (see NX for details).

```
> restart : with( Optimization ) : with( CAD:-NX ) :
```

Read existing design parameters in Maple

```
> pars := GetParameterNames( )
> heightConeCurr := GetParameterValue( "heightCone" ) · GetParameterUnits( "heightCone" )
```

$$heightConeCurr := 50.00\ \text{mm} \tag{2.1}$$

```
> lenCentralCurr := GetParameterValue( "lenCentral" ) · GetParameterUnits( "lenCentral" )
```

$$lenCentralCurr := 60.00\ \text{mm} \tag{2.2}$$

```
> radSphereCurr := GetParameterValue( "radSphere" ) · GetParameterUnits( "radSphere" )
```

$$radSphereCurr := 15.00\ \text{mm} \tag{2.3}$$

Define expressions to calculate the volume and surface area of the cannister

```
> volCanister := ( heightCone, lenCentral, radSphere ) → evalf( (π radSphere^2 heightCone)/3 + π radSphere^2 lenCentral
      + (2 π radSphere^3)/3 ) :
> vol := volCanister( heightConeCurr, lenCentralCurr, radSphereCurr )
```

$$vol := 61261.05675\ \text{mm}^3 \tag{3.1}$$

```
> surfaceCanister := ( heightCone, lenCentral, radSphere ) → evalf( 2 π radSphere^2 + 2 π radSphere lenCentral
      + π radSphere sqrt( radSphere^2 + heightCone^2 ) ) :
> surfaceCanister( heightConeCurr, lenCentralCurr, radSphereCurr )
```

$$9528.522740\ \text{mm}^2 \tag{3.2}$$

Design Optimization

Minimize the surface area while maintaining the volume

```
> optValues := Minimize( surfaceCanister( heightConeNew, lenCentralNew, radSphereNew ),
      { volCanister( heightConeNew, lenCentralNew, radSphereNew ) = vol/mm^3 }, assume = nonnegative )
```

$$optValues := [7660.22, [heightConeNew = 21.46, lenCentralNew = 10.73, radSphereNew = 23.99]] \tag{4.1}$$

Export new design parameters into NX

```
> assign( optValues[2] ) :
> SetParameterValue( "heightCone", heightConeNew ) :
> SetParameterValue( "lenCentral", lenCentralNew ) :
> SetParameterValue( "radSphere", radSphereNew ) :
```